

# Come non atterrare a Lake Tahoe!

## - Note

### - Autore

Claudio Marsan  
Liceo Cantonale di Mendrisio  
Via Agostino Maspoli  
CH-6850 Mendrisio (Switzerland)  
e-mail: claudio.marsan@liceomendrisio.ch

### - Versione

Versione 3.0, 14 marzo 2003  
Testato con: Maple V Release 6.02 for Windows 2000

### - Bibliografia

RICHARD BARSHINGER  
How Not to Land at Lake Tahoe!  
American Mathematical Monthly, May 1992, pp. 453-455

## - Il problema

Consideriamo un aeroplano che si appresta ad eseguire la manovra di atterraggio seguendo una curva che può essere descritta da una cubica. L'aeroplano si trova ad un'altezza  $H$  e ad una distanza orizzontale  $L$  dall'aeroporto. Vogliamo costruire un modello per la fase di atterraggio supponendo che la componente orizzontale della velocità sia costante durante tutta la fase d'atterraggio (ipotesi poco realistica!) e che la componente verticale dell'accelerazione sia sempre, in valore assoluto, minore di  $k$  e molto più piccola dell'accelerazione di gravità.

## - Risoluzione del problema

```
[ > restart:
[  $H$  e  $L$  sono costanti.
[ > constants := constants, H, L:
[  $y(x)$  sia la funzione che descrive la traiettoria seguita dall'aereo nella manovra di atterraggio.
[ >  $y := x \rightarrow a*x^3 + b*x^2 + c*x + d;$ 
[  $y := x \rightarrow a x^3 + b x^2 + c x + d$ 
[  $dy$  sia la derivata della funzione che descrive la traiettoria.
[ >  $dy := unapply(diff(y(x), x), x);$ 
[  $dy := x \rightarrow 3 a x^2 + 2 b x + c$ 
```

Come origine del sistema di riferimento scegliamo l'aeroporto. Si pongono le seguenti 4 condizioni:

[ - Posizione dell'aereo quando tocca terra.

```
> eq1 := y(0) = 0;
```

$$eq1 := d = 0$$

- Atterraggio dolce (no crash!).

```
> eq2 := dy(0) = 0;
```

$$eq2 := c = 0$$

- Posizione dell'aereo all'inizio della manovra d'atterraggio.

```
> eq3 := y(-L) = H;
```

$$eq3 := -a L^3 + b L^2 - c L + d = H$$

- La manovra di atterraggio inizia dolcemente (non in picchiata!).

```
> eq4 := dy(-L) = 0;
```

$$eq4 := 3 a L^2 - 2 b L + c = 0$$

Risolviendo il sistema determinato dalle 4 condizioni si ottengono i valori dei parametri  $a, b, c, d$ .

```
> solve({eq1, eq2, eq3, eq4}, {a,b,c,d});
```

$$\{c = 0, d = 0, a = 2 \frac{H}{L^3}, b = 3 \frac{H}{L^2}\}$$

Ecco l'equazione della traiettoria, dipendente da  $H$  e  $L$ .

```
> assign(%): 'y(x)' = y(x);
```

```
>
```

$$y(x) = 2 \frac{H x^3}{L^3} + \frac{3 H x^2}{L^2}$$

I calcoli seguenti servono per determinare  $v_y$ , la componente verticale della velocità, tenendo conto che la componente orizzontale della velocità è considerata uguale alla costante  $U$ .

```
> v_y := diff(y(x(t)), t);
```

```
> v_y := factor(v_y)/diff(x(t), t)*U;
```

$$v_y := 6 \frac{H x(t) (x(t) + L) U}{L^3}$$

I calcoli seguenti servono per determinare la componente verticale  $a_y$  dell'accelerazione, tenendo conto che la componente orizzontale dell'accelerazione è nulla.

```
> a_y := diff(v_y, t);
```

```
> a_y := factor(a_y)/diff(x(t), t)*U;
```

$$a_y := 6 \frac{H U^2 (2 x(t) + L)}{L^3}$$

Per comodità sostituiamo  $x(t)$  con  $x$  nelle formule per  $v_y$  e per  $a_y$

```
> v_y := subs(x(t)=x, v_y);
```

```
> a_y := subs(x(t)=x, a_y);
```

$$v_y := 6 \frac{H x (x + L) U}{L^3}$$

$$a_y := 6 \frac{H U^2 (2 x + L)}{L^3}$$

La componente verticale dell'accelerazione all'inizio della fase di atterraggio.

```
> a_iniziale := subs(x=-L, a_y);
```

$$a_{iniziale} := -6 \frac{H U^2}{L^2}$$

La componente verticale dell'accelerazione alla fine della fase di atterraggio.

```
> a_finale := subs(x=0, a_y);
```

$$a_{finale} := 6 \frac{H U^2}{L^2}$$

La componente verticale dell'accelerazione massima in valore assoluto.

```
> a_max := max(abs(a_iniziale), abs(a_finale));
```

$$a_{max} := 6 \left| \frac{H U^2}{L^2} \right|$$

La condizione che deve essere soddisfatta.

```
> a_max <= k;
```

$$6 \left| \frac{H U^2}{L^2} \right| \leq k$$

Per gli esempi che seguono è necessario indicare a Maple che le unità di misura per lo spazio e per il tempo sono grandezze positive.

```
> assume(piedi>0): assume(metri>0): assume(miglia>0):  
  assume(secondi>0): assume(ore>0):
```

## - Esempio 1

Consideriamo un volo transatlantico con un (pesante) Jumbo Jet, per esempio tra LAX (Los Angeles) e JFK (New York).  $U$  e  $H$  sono grandi,  $k$  è piccolo e quindi  $L$  deve essere grande: infatti l'aereo inizia la fase di atterraggio a Scranton (PA), circa 130 miglia da New York.

```
> k_max := evalf(subs({H=37000*piedi, U=600*miglia/ore,  
  L=130*miglia}, a_max));
```

$$k_{max} := .4728994083 \cdot 10^7 \frac{\text{piedi} \sim}{\text{ore} \sim^2}$$

Valore massimo di  $k$ , in piedi al secondo al secondo.

```
> `valore massimo di k` = evalf(subs(ore=3600*secondi, k_max))  
  ;
```

$$\text{valore massimo di } k = .3648915187 \frac{\text{piedi} \sim}{\text{secondi} \sim^2}$$

Valore massimo di  $k$ , in metri al secondo al secondo.

```
> simplify(subs(piedi=.3048*metri, %));
```

$$\text{valore massimo di } k = .1112189349 \frac{\text{metri} \sim}{\text{secondi} \sim^2}$$

## - Esempio 2

Consideriamo ora un piccolo aereo che viaggia tra San Francisco e South Lake Tahoe (località che si trova ad un'altezza di 6228 piedi sul livello del mare, sulla Sierra Nevada, al confine tra la California e il Nevada). La pista di atterraggio è corta,  $L$  e  $k$  sono piccoli e  $H$  è relativamente grande (ci sono delle montagne). Allora  $U$  deve essere mantenuta bassa.

```
> k_max := simplify(subs({H=11000*piedi, U=175*miglia/ore,
L=20*miglia}, a_max));
```

$$k_{max} := 5053125 \frac{\text{piedi}}{\text{ore}^2}$$

Valore massimo di  $k$ , in piedi al secondo al secondo.

```
> `valore massimo di k` = evalf(subs(ore=3600*secondi, k_max))
;
```

$$\text{valore massimo di } k = .3899016204 \frac{\text{piedi}}{\text{secondi}^2}$$

Valore massimo di  $k$ , in metri al secondo al secondo

```
> simplify(subs(piedi=.3048*metri, %));
```

$$\text{valore massimo di } k = .1188420139 \frac{\text{metri}}{\text{secondi}^2}$$